
บทที่   Â
ฟงกชัน (Function)

จดุประสงคเชิงพฤติกรรม
1. เขาใจพื้นฐานการฟงกชันใชงาน
2. ใชอางองิเมื่อมีความตองการ  แตนึกรูปแบบฟงกชันไมออก
3. ใชอางองิเมื่อตองการหาฟงก็ชัน ไปแกปญหาในการเขียนโปรแกรม

หวัขอในบทเรียน
ฟงกชันทั้งหมด 194 ฟงกชัน
จาก ฟงกชัน AADD () จนถึง ฟงกชัน YEAR()
ตวัอยางโปรแกรมเกี่ยวกับฟงกชัน  227 ตัวอยาง

หมายเหตุ
มฟีงกชันใชงานมากมายที่ Clipper มใีหบรกิาร มีฟงกชันหลายตัวที่มีหนาที่คลายกันทั้งในฟงกชัน หรือค ํา

สัง่เอง กเ็พือ่สรางทางเลือกใหกับผูใชในการเขียนโปรแกรม และหลายคนมีพื้นฐานมาจากภาษาอื่น ก็สามารถเลือกใช
ฟงกชนัทีต่นคุนเคยไดโดยงาย บทนี้ตองการเปนแหลงอางอิง เพราะมีนักเรียน หรือนักพัฒนานอยคน ที่จะจํ าฟงกชัน
ทัง้หมดไดครบ หรือจ ํารูปแบบการใชไดทั้งหมด .. แหลงอางอิงจึงมีความสํ าคัญ

บทที่ 3   ฟงกชัน (Function)
ฟงกชนั คอืการประมวลผลกับคาที่สงใหฟงกชันดํ าเนินการ  แลวคืนคาที่ผานการประมวลผลกลับมา   หรือ

การหาคาโดยใชฟงกชัน    นั่นหมายความวา ฟงกชันมีหนาที่คืนคาใหกับโปรแกรม  แลวโปรแกรมสามารถน ําคาไปใช
ไดตามตองการ  แตจะคือคามาในรูปแบบใด ขึ้นอยูกับหนาที่ของฟงกชันนั้น ๆ

&  3.1 AADD()  10

! รูปแบบ
AADD(<ชดุอาเรยเปาหมาย>,<นิพจนที่เพิ่มเขาไป>) --> คาของนิพจน
P หนาที่
ฟงกชนันี้จะทํ าหนาที่ เพิ่มสมาชิกอีก 1 ตัวเขาไปตอทายอาเรยชุดเดิมโดยจะเพิ่มคาเขาไปตออาเรยตัวสุดทาย แลวสงคา
คนื ซึง่เปนคาของนิพจนที่เพิ่มเขาไป
: ตัวอยางที่ 3.1
        AR := {}                // อาเรยวางเปลา
        AADD(AR,'BAT')          // {'BAT'}
        AADD(AR,'RAT')          // {'BAT','RAT'}
        AADD(AR,2500)           // {'BAT','RAT',2500}
        FOR I = 1 TO 3
           ?? AR[I]             // BATRAT      2500
        NEXT
: ตัวอยางที่ 3.2
        AR := {}
        TEST := 'ABC'
        AADD(AR,'A')
        AADD(AR,5)
        ? AADD(AR,'XYZ')        // XYZ
        AADD(AR,{'B',TEST,23})
        ? AR[1]                 // A
        ? AR[2]                 //          5
        ? AR[3]                 // XYZ
        ? AR[4,1]               // B
        ? AR[4,2]               // ABC
        ? AR[4,3]               //         23
        ? '====='               // =====

10  นิพนธ กิตติปภัสสร และ อนันต  อุตตะมะ,  อางแลวในเชิงอรรถที่ ( 1 ),  หนา 135

&  3.2 ABS()
! รูปแบบ
ABS(<นพิจนจ ํานวนเลข>) --> จํ านวนเลขบวกหรือศูนย
P หนาที่
ฟงกชันนี้จะทํ าหนาที่ แปลงตัวแปรตัวเลขเปนคาสัมบูรณ
: ตัวอยางที่ 3.3
        TEST1 := 10
        TEST2 := 12
        ? (TEST1 - TEST2)       //         -2
        ? ABS(TEST1 - TEST2)    //          2
        ? ABS(TEST1 - 100)      //         90
        ? ABS(3 - 7)            //          4
        ? ABS(0)                //          0
        ? ABS(5)                //          5
        ? ABS(22 / 7)           //          3.14
        ? ABS(2.555 - 3.4444)   //          0.8894

&  3.3 ACHOICE()
! รูปแบบ
ACHOICE(<มมุบน>, <มุมซาย>, <มุมลาง>, <มุมขวา>,
        <ชดุอาเรยแสดงเมน>ู) --> จํ านวนเลขของตัวเลือกที่ถูกเลือก
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สั่งใหนํ า ARRAY มาเปนเมนูแบบ PULLDOWN MENU
: ตัวอยางที่ 3.4
        CLS
        DO WHILE .T.
           OPT := ACHOICE(5,10,7,30,{'ADD','DELETE';
                  ,'CHANGE','PRINT','QUIT'})
           OPTPROGRAM := {'ADDPRO','DELPRO','CHGPRO','PRTPRO'}
           DO CASE
              CASE OPT = 1
                     DO ADDPRO
              CASE OPT = 2

                     DO DELPRO
              CASE OPT = 3
                     DO CHGPRO
              CASE OPT = 4
                     DO PRTPRO
              CASE OPT = 5
                     EXIT
           ENDCASE
        ENDDO
: ตัวอยางที่ 3.5
        CLS
        OPTPROGRAM := {'ADDPRO','DELPRO','CHGPRO','PRTPRO'}
        DO WHILE .T.
           OPT := ACHOICE(5,10,7,30,{'ADD','DELETE';
                  ,'CHANGE','PRINT','QUIT'})
           IF OPT = 5 .OR. LASTKEY() = 27
              EXIT
           ENDIF
           ? (OPTPROGRAM[OPT])
           DO CASE
              CASE OPT = 1;   DO ADDPRO
              CASE OPT = 2;   DO DELPRO
              CASE OPT = 3;   DO CHGPRO
              CASE OPT = 4;   DO PRTPRO
           ENDCASE
        ENDDO
: ตัวอยางที่ 3.6
        CLS
        OPTMENU    := {'ADD','DELETE','CHANGE','PRINT','QUIT'}
        OPTPROGRAM := {'ADDPRO','DELPRO','CHGPRO','PRTPRO'}
        DO WHILE .T.
           OPT := ACHOICE(5,10,7,30,OPTMENU)
           DO CASE

              CASE OPT = 5 .OR. LASTKEY() =27 ; EXIT
              CASE OPT = 1;   DO ADDPRO
              CASE OPT = 2;   DO DELPRO
              CASE OPT = 3;   DO CHGPRO
              CASE OPT = 4;   DO PRTPRO
           ENDCASE
        ENDDO

&  3.4 ACLONE()
! รูปแบบ
ACLONE(<ชดุอาเรยตนแบบ>) --> ชุดอาเรยชุดใหม
P หนาที่
ฟงกชนันี้จะท ําหนาที่ คัดลอกอาเรยทั้งชุดไปสรางเปนชุดใหม
: ตัวอยางที่ 3.7
        OLD := {'ABC',12,2.5}
        NEW := OLD                      // 2 ตวัแปร แตขอมูลชุดเดียว
        OLD[1] := 'DEF'
        ? OLD[1], OLD[2], OLD[3]        // DEF  12   2.5
        ? NEW[1], NEW[2], NEW[3]        // DEF  12   2.5
: ตัวอยางที่ 3.8
        OLD := {'ABC',12,2.5}
        NEW := ACLONE(OLD)              // เก็บขอมูลไว 2 ชุด
        OLD[1] := 'DEF'
        ? OLD[1], OLD[2], OLD[3]        // DEF  12   2.5
        ? NEW[1], NEW[2], NEW[3]        // ABC  12   2.5

&  3.5 ACOPY()
! รูปแบบ
ACOPY(<ชดุอาเรยตนแบบ>, <ชุดอาเรยเปาหมาย>,
      [<เริม่ทีต่ ําแหนง>], [<จํ านวนตํ าแหนง>],
      [<ต ําแหนงปลายทาง>]) --> ชุดอาเรยปลาย
P หนาที่
ฟงกชนันี้จะท ําหนาที่ คัดลอกอาเรยไปทับอาเรยชุดใหม
: ตัวอยางที่ 3.9

        OLD := {'A','B','C'}
        NEW := {1,2,3}
        ACOPY(OLD,NEW,1,1,1)    // OLD ตวัแรกตัวเดียวทับ NEW ตัวเดียว
        ? NEW[1],NEW[2],NEW[3]  // A   2   3
        ACOPY(OLD,NEW,1,1,2)    // OLD ตวัแรกทับ NEW 2 ชุด ไปต ําแหนงที่ 2
        ? NEW[1],NEW[2],NEW[3]  // A   A   3
        ACOPY(OLD,NEW,1,2,1)    // OLD จากตัวแรกไป 2 ตัว ไปต ําแหนงที่ 1
        ? NEW[1],NEW[2],NEW[3]  // A   B   3
        ACOPY(OLD,NEW,2,2,1)   // OLD จากตัวที ่2 ไป 2 ตัว ไปต ําแหนงที่ 1
        ? NEW[1],NEW[2],NEW[3]  // B   C   3
        ACOPY(OLD,NEW,3,1,1)    // OLD จากตัวที ่3 ไป 1 ตัว ไปต ําแหนงที่ 1
        ? NEW[1],NEW[2],NEW[3] // C   C   3
        ACOPY(OLD,NEW,1,1,3)    // OLD จากตัวที ่1 ไปต ําแหนงที่ 3
        ? NEW[1],NEW[2],NEW[3]  // C   C   A

&  3.6 ADEL()
! รูปแบบ
ADEL(<ชดุอาเรยเปาหมาย>, <ตํ าแหนงที่>) --> ชุดอาเรยปลายทาง
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ลบอาเรยบางตัวออกจากชุดอาเรยที่มีอยู
: ตัวอยางที่ 3.10
        OLD := {'A','B','C'}
        NEW := {1,2,3}
        ACOPY(OLD,NEW,1,1,3)
        ? NEW[1],NEW[2],NEW[3]                  //  1  2  A
        ADEL(OLD,2)
        ADEL(NEW,3)
        ? ; FOR I = 1 TO 3 ; ?? OLD[I] ; NEXT   //  A  C  NIL
        ? ; FOR I = 1 TO 3 ; ?? NEW[I] ; NEXT   //  1  2  NIL

&  3.7 ADIR()
! รูปแบบ
ADIR([<ชือ่แฟมที่ตองการอานเขามา>],
     [<ชดุอาเรยที่รับชื่อแฟม>],     [<ชุดอาเรยที่รับขนาด>],
     [<ชดุอาเรยที่รับวันที>่],     [<ชุดอาเรยที่รับเวลา>],
     [<ชดุอาเรยทีร่ับลักษณะเฉพาะ>]) --> จํ านวนแฟมที่รับเขามา
P หนาที่
ฟงกชนันี้จะท ําหนาที่ น ําขอมูลแฟมเชน ชื่อแฟม หรือขนาด เก็บลงอาเรย
: ตัวอยางที่ 3.11
        LOCAL FPRG[20]               // จองเนื้อที่เปนอาเรยให FPRG
        LOCAL FPRGTEST := ARRAY(20)  // จองเนื้อที่เปนอาเรยอีกแบบหนึ่ง
        ADIR("*.PRG",FPRG)
        FOR I = 1 TO 3               // X.PRG
            ? FPRG[I]                // Y.PRG
        NEXT                         // Z.PRG
: ตัวอยางที่ 3.12
        LOCAL FINDDIR := "C:\WINDOWS\*.EXE"
        LOCAL FDIRS[ADIR(FINDDIR)]   // ท ําใหจองเนื้อที่ใหอาเรยไดถูกตอง
        ADIR(FINDDIR,FDIRS)
        AEVAL(FDIRS,{|ELEMENT| QOUT(ELEMENT)}) // QOUT ใหพิมพอาเรยได
: ตัวอยางที่ 3.13
        LOCAL FINDDIR := "C:\WINDOWS\*.EXE"
        LOCAL FDIRS[ADIR(FINDDIR)] // ท ําใหจองเนื้อที่ใหอาเรยไดถูกตอง
        LOCAL FSIZES [ADIR(FINDDIR)]
        LOCAL FDATES [ADIR(FINDDIR)]
        LOCAL FTIMES [ADIR(FINDDIR)]
        LOCAL FATTRIBUTES [ADIR(FINDDIR)]
        ADIR(FINDDIR,FDIRS,FSIZES,FDATES,FTIMES,FATTRIBUTES)
        AEVAL(FDIRS ,{|ELEMENT| QOUT(ELEMENT)}) ; INKEY(0)
        AEVAL(FSIZES,{|ELEMENT| QOUT(ELEMENT)}) ; INKEY(0)
        AEVAL(FDATES,{|ELEMENT| QOUT(ELEMENT)}) ; INKEY(0)
        AEVAL(FTIMES,{|ELEMENT| QOUT(ELEMENT)}) ; INKEY(0)
        AEVAL(FATTRIBUTES,{|ELEMENT| QOUT(ELEMENT)}) ; INKEY(0)

: ตัวอยางที่ 3.14
        LOCAL FINDDIR
        ACCEPT TO FINDDIR
        FDIRS  = ARRAY(ADIR(FINDDIR))
        FSIZES  = ARRAY(ADIR(FINDDIR))
        FDATES  = ARRAY(ADIR(FINDDIR))
        FTIMES  = ARRAY(ADIR(FINDDIR))
        FATTRIBUTES = ARRAY(ADIR(FINDDIR))
        ADIR(FINDDIR,FDIRS,FSIZES,FDATES,FTIMES,FATTRIBUTES)
        FOR I = 1 TO ADIR(FINDDIR)
           ? FDIRS[I],FSIZES[I],FDATES[I],FTIMES[I],FATTRIBUTES[I]
        NEXT

&  3.8 AEVAL()
! รูปแบบ
AEVAL(<ชดุอาเรย>, <นิพจนที่น ําอาเรยไปใช>,
      [<เริม่ทีต่ ําแหนง>], [<จํ านวนตํ าแหนง>]) --> คาจากนิพจน
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สั่งประมวลผลกับอาเรยดวยชุดคํ าสั่งแบบบล็อก(BLOCK)
: ตัวอยางที่ 3.15
        LOCAL ELE    := "{|ELEMENT| QOUT(ELEMENT)}"
        LOCAL ELEMIN :="{|ELEMENT| IF(ELEMENT<MIN,MIN:=ELEMENT,) }"
        LOCAL ELECNT :="{|ELEMENT| CNT:=CNT+1}"
        // LOCAL ELECNT:="{|ENIL|IF(ENIL<>NIL,CNT:=CNT+1)}"เฉพาะที ่!= NIL
        LOCAL ELESUM :="{|ELEMENT| SUM := SUM + ELEMENT}"
        MAX = 0
        MIN = 9999
        CNT = 0
        SUM = 0
        X := {40,12,8,5,21}
        Y := ACLONE(X)
        AEVAL( X , {|ELEMENT| QOUT(ELEMENT)})  ; INKEY(0)
        AEVAL( X , &ELE)                       ; INKEY(0)
        AEVAL( X , {|ELEMENT| IF(ELEMENT > MAX ,MAX := ELEMENT,)})

        AEVAL( X , &ELEMIN)
        AEVAL( X , &ELECNT)
        AEVAL( X , &ELESUM)
        ? MAX , MIN , CNT , SUM/CNT

&  3.9 AFIELDS()
! รูปแบบ
AFIELDS([<ชดุอาเรยเก็บชื่อฟลด>], [<ชุดอาเรยเก็บแบบ>],
        [<ชดุอาเรยเกบ็ความกวาง>], [<ชุดอาเรยเก็บจํ านวนทศนิยม>])
        --> จ ํานวนฟลดของแฟมปจจุบัน
P หนาที่
ฟงกชันนี้จะท ําหนาที่ ใสโครงสรางแฟมปจจุบันลงไปใน ARRAY
: ตัวอยางที่ 3.16
        USE FILEA
        AFLDNAME := ARRAY(AFIELDS())
        AFLDTYPE := ARRAY(AFIELDS())
        AFLDWIDTH:= ARRAY(AFIELDS())
        AFLDDEC  := ARRAY(AFIELDS())
        AFIELDS(AFLDNAME,AFLDTYPE,AFLDWIDTH,AFLDDEC)
        FOR I = 1 TO AFIELDS()                  // FIELD1 C  10
            ?  AFLDNAME[I],AFLDTYPE[I]          // FIELD2 N  10  0
            ?? STR(AFLDWIDTH[I],10)             // FIELD3 N  10  0
            ?? IF(AFLDTYPE[I]='N',STR(AFLDDEC[I],10),'')
        NEXT

&  3.10 AFILL()
! รูปแบบ
AFILL(<ชดุอาเรยเปาหมาย>, <คาที่จะนํ าเขาชุดอาเรย>,
      [<เริม่ทีต่ ําแหนง>], [<จํ านวนตํ าแหนง>]) --> ชุดอาเรยเปาหมาย
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ใสคาที่กํ าหนดลงไปใน ARRAY โดยรูปแบบขอมูลแตละครั้งก ําหนดไดเพียงแบบเดียว
: ตัวอยางที่ 3.17
        LOCAL  X[5]     // NIL NIL NIL NIL NIL
        LOCAL  Y[5]     // NIL NIL NIL NIL NIL
        AFILL(X,12)     // น ําเลข 12 ใสในอาเรยทุกตัว
        FOR I = 1 TO 5 ; ?X[I] ; NEXT   // 12 12 12 12 12

        AFILL(X,5,2,3)    // ใสเลข 5 เริ่มที่ตํ าแหนง 2 ไป 3 ตัว
        FOR I = 1 TO 5 ; ?X[I] ; NEXT   // 12  5  5  5 12
        AFILL(X,.T.,3,2)  // ใส .T. เริ่มที่ตํ าแหนง 3 ไป 2 ตัว
        FOR I = 1 TO 5 ; ?X[I] ; NEXT   // 12  5 .T. .T. 12

&  3.11 AINS()
! รูปแบบ
AINS(<ชดุอาเรยเปาหมาย>, <ตํ าแหนงที่>) --> ATARGET
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ แทรกคาวาง(NIL) ลงไปในอาเรย ท ําใหตัวสุดทายหายไปเสมอ
: ตัวอยางที่ 3.18
        X := {1,2,3,4,5}
        AINS(X,2)                       // แทรกเขาไปในตํ าแหนงที่ 2
        FOR I = 1 TO 5 ; ?X[I] ; NEXT   // 1  NIL  2  3  4
        AINS(X,4)                       // แทรกเขาไปในตํ าแหนงที่ 4
        FOR I = 1 TO 5 ; ?X[I] ; NEXT   // 1  NIL  2  NIL  3

&  3.12 ALERT()
! รูปแบบ
ALERT(<ขอความ> [, <ชุดอาเรยตัวเลือก>]) --> คาตัวเลือก
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ การรับคาจากตามตัวเลือกในแบบไดอาลอกบล็อก (DIALOG BOX)
: ตัวอยางที่ 3.19
        X = {'A','B','C','D'}           // ตองไมเกิน 4 คา
        OPT = ALERT('TEST ALERT',X)
        ? OPT                           // ผลคือ 1 ถึง 4 เทานั้น
        HEADOPT = 'ตดัสินใจส ําหรับรายงานนี้'
        TXTOPT = {'SCREEN','PRINTER','EDIT OPTION','CANCEL'}
        OPT = ALERT(HEADOPT,TXTOPT)
        DO CASE
           CASE OPT = 1 ; ? 'พมิพรายงานทางจอภาพ'
           CASE OPT = 2 ; ? 'พมิพรายงานทางเครื่องพิมพ'

           CASE OPT = 3 ; ? 'กลับไปแกไขตัวเลือก'
           CASE OPT = 4 ; ? 'ยกเลิก และกลับเมน'ู
           OTHERWISE    ; ? 'ทานไมเลือกปุมใดเลย รูตัวรึเปลา'
        ENDCASE
: ตัวอยางที่ 3.20
        HEADOPT = 'ตดัสนิใจส ําหรับรายงานนี้ และไมอนุญาติใหกด ESC'
        TXTOPT = {'SCREEN','PRINTER','EDIT OPTION','CANCEL'}
        OPT = 4
        DO WHILE .T.
           OPT = ALERT(HEADOPT,TXTOPT)
           DO CASE
              CASE OPT = 1 ; ? 'พมิพรายงานทางจอภาพ'
              CASE OPT = 2 ; ? 'พมิพรายงานทางเครื่องพิมพ'
              CASE OPT = 3 ; ? 'กลับไปแกไขตัวเลือก'
              CASE OPT = 4 ; ? 'ยกเลิก และกลับเมน'ู; EXIT
              OTHERWISE
                ALERT('ทานไมเลือกปุมใดเลย รูตัวรึเปลา')
           ENDCASE
        ENDDO

&  3.13 ALIAS()
! รูปแบบ
ALIAS([<พืน้ทีท่ ํางานที่>]) --> สมนาม
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สงชื่อพื้นที่ท ํางานปจจุบัน
: ตัวอยางที่ 3.21
        SELE 1
        USE FILEA
        ? ALIAS()                  // FILEA
        SELE 2
        USE FILEB ALIAS AREANAME2
        ? ALIAS()                  // AREANAME2
        SELE 3
        USE FILEC ALIAS AREANAME3  // ชื่อ ALIAS ตองไมเกิน 10

        ? ALIAS()                  // AREANAME3
        SELE AREANAME2
        ? ALIAS()                  // AREANAME2

&  3.14 ALLTRIM()
! รูปแบบ
ALLTRIM(<ขอความ>) --> ขอความที่ผานการตัดชองวาง
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ น ําคาวางออกจากตัวแปรตัวอักษร ทั้งดานซาย และดานขวา
: ตัวอยางที่ 3.22
        X = "ABC "
        Y = " ABC"
        Z = " ABC "
        ? LEN(ALLTRIM(X))                       //          3
        ? ALLTRIM(X)+ALLTRIM(Y)                 // ABCABC
        ? LEN(LTRIM(X))                         //          4
        ? LEN(RTRIM(X))                         //          3
        ? LEN(LTRIM(Z))                         //          4
        ? LEN(RTRIM(Z))                         //          4
        ? LEN(ALLTRIM(Z))                       //          3
        ? ALLTRIM(X)+ALLTRIM(Y)+ALLTRIM(Z)  // ABCABCABC
        ? RTRIM(X)+RTRIM(Y)+RTRIM(Z)            // ABC ABC ABC
        ? LTRIM(X)+LTRIM(Y)+LTRIM(Z)            // ABC ABCABC

&  3.15 ARRAY()
! รูปแบบ
ARRAY(<จ ํานวนสมาชกิ> [, <จํ านวนสมาชิก>...]) --> ชุดอาเรย
P หนาที่
ฟงกชนันี้จะท ําหนาที่ กํ าหนดใหมีการใชตัวแปร ARRAY
: ตัวอยางที่ 3.23
        AR := ARRAY(3)          // การก ําหนดอาเรยมิติเดียวมาใช
        AR := {NIL , NIL , NIL} // หรือก ําหนดแบบใช {}
        FOR I = 1 TO 3
            ACCEPT TO AR[I]

        NEXT
        FOR I = 1 TO 3
            ? AR[I]
        NEXT
: ตัวอยางที่ 3.24
        AR := ARRAY(2,3)        // การก ําหนดอาเรยหลายมิติมาใช
        AR := {{NIL,NIL,NIL},{NIL,NIL,NIL}}
        AR := {ARRAY(3),ARRAY(3)}
        // น ําขอมูลจากโครงสรางแฟมเก็บในอาเรยหลายมิติ
        USE FILEA
        FLDAR := ARRAY(4,AFIELDS())
        AR1 := FLDAR[1]
        AR2 := FLDAR[2]
        AR3 := FLDAR[3]
        AR4 := FLDAR[4]
        AFIELDS(AR1,AR2,AR3,AR4)
        FOR I = 1 TO AFIELDS()
            ? FLDAR[1,I],FLDAR[2,I],FLDAR[3,I],FLDAR[4,I]
        NEXT
        FOR I = 1 TO AFIELDS()
            ?
            FOR J = 1 TO 4
                ?? FLDAR[J,I]
            NEXT
        NEXT
        AEVAL(FLDAR,{|ELE|QOUT(ELE[1],ELE[2])})

&  3.16 ASC()
! รูปแบบ
ASC(<ตวัอกัษร>) --> เลขล ําดับในตารางแอสกี (ASCII)
P หนาที่
ฟงกชนันี้จะทํ าหนาที่ แปลงคาใหเปนรหัส ASCII
: ตัวอยางที่ 3.25
        ? ASC("A")              //         65
        ? ASC("A")             //         97
        ? ASC("ABCDEF")         //         97
        ? ASC("Z")              //         90

        ? ASC("Z123")           //         90
        ? ASC("Z")              //        122
        ? ASC("Z") - ASC("A")   //         25
        ? ASC("Z") - ASC("A")   //         25
        FOR I = ASC("A") TO ASC("Z")
            ? I                 //    97 - 122
        NEXT

&  3.17 ASCAN()
! รูปแบบ
ASCAN(<ชดุอาเรยเปาหมาย>, <ขอความที่ตองการหา>,
      [<เริม่ทีต่ ําแหนง>], [<จํ านวนตํ าแหนง>]) --> เลขแสดงตํ าแหนงที่พบ
P หนาที่
ฟงกชนันี้จะท ําหนาที่ คนหาคาที่ตองการในอาเรย
: ตัวอยางที่ 3.26
        AR := {'LION','TIGER','STAR','COCONUT','KEYBOARD'}
        ? ASCAN(AR,'LION')              //         1
        ? AR[ASCAN(AR,'LION')]          //    LION
        ? ASCAN(AR,'LION')              //         0

&  3.18 ASIZE()
! รูปแบบ
ASIZE(<ชดุอาเรยเปาหมาย>, <เลขกํ าหนดจํ านวนสมาชิก>) --> ชุดอาเรยเปาหมาย
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สั่งกํ าหนดขนาดของชุดอาเรย ซึ่งเพิ่มหรือลดจากเดิมได
: ตัวอยางที่ 3.27
        AR  := {'A',1}
        CNT := 0
        AEVAL(AR,{|ELEMENT| CNT:=CNT+1})  ; ? CNT    // 2
        ASIZE(AR,4)             // A   1   NIL   NIL
        CNT := 0
        AEVAL(AR,{|ELEMENT| CNT:=CNT+1})  ; ? CNT    // 4
        ASIZE(AR,3)             // A   1   NIL
        CNT := 0

PDF created with FinePrint pdfFactory Pro trial version http://www.pdffactory.com

http://www.pdffactory.com


                                                                    บทที่ 3  :  ฟงกชัน (Function)                                                               87

        AEVAL(AR,{|ELEMENT| CNT:=CNT+1})  ; ? CNT    // 3
        ASIZE(AR,CNT-1)         // A   1
        CNT := 0
        AEVAL(AR,{|ELEMENT| CNT:=CNT+1})  ; ? CNT    // 2
        FOR I = 1 TO CNT
            ? AR[I]
        NEXT

&  3.19 ASORT()
! รูปแบบ
ASORT(<ชดุอาเรยเปาหมาย>, [<เริ่มที่ตํ าแหนง>],
      [<จ ํานวนต ําแหนง>], [<ตรรกเลือกแบบในการจัดเรียง>]) --> ชุดอาเรยเปาหมาย
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สั่งจัดเรียงสมาชิกในอาเรย
: ตัวอยางที่ 3.28
        AR := {5,3,1,2,4}
        X  := ACLONE(AR)
        ASORT(AR)                  // จดัเรยีงสมาชิกทุกตัวจากนอยไปมาก
        ASORT(AR,,,{|X,Y| X > Y})  // จดัเรยีงสมาชิกทุกตัวจากมากไปนอย
        ASORT(AR,2,3)              // จดัเรยีงตัวที ่2 ถึง 4 จากนอยไปมาก
        AEVAL(AR,{|ELE|QOUT(ELE)}) // 5 2 3 4 1
: ตัวอยางที่ 3.29
        AR := ARRAY(4,2)
        AR[1,1] := 'DANG'  ; AR[1,2] := 12
        AR[2,1] := 'BOY'   ; AR[2,2] := 24
        AR[3,1] := 'JOJO'  ; AR[3,2] := 11
        AR[4,1] := 'ANON'  ; AR[4,2] := 18
        ASORT(AR)                   // จดัเรยีงตามชื่อ จากนอยไปมาก
        ASORT(AR,,,{|X,Y|X[2]<Y[2]}) // จดัเรยีงตามอายุจากนอยไปมาก
        AEVAL(AR,{|ELE|QOUT(ELE[1],ELE[2])})
        ASORT(AR,,,{|X,Y|X[2]>Y[2]}) // จดัเรยีงตามอายุจากมากไปนอย
        AEVAL(AR,{|ELE|QOUT(ELE[1],ELE[2])}) // น ําอาเรยมาพิมพ

&  3.20 AT()
! รูปแบบ
AT(<ตวัอกัษรทีต่องการหา>, <ขอความเปาหมาย>) --> ตํ าแหนงที่พบ
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สงตํ าแหนงของขอความยอย ในขอความที่ตองการ
: ตัวอยางที่ 3.30
        ? AT('A','IJUSTCALLTOSAYILOVEYOU')      // 7
        TXT := 'MR.SILICON VLSI'
        ? SUBSTR(TXT,AT('.',TXT)+1)             // SILICON VLSI
        ? SUBSTR(TXT,1,AT(' ',TXT)-1)           // MR.SILICON
        X := SUBSTR(TXT,1,AT(' ',TXT)-1)
        ? SUBSTR(X,AT('.',X)+1)                 // SILICON

&  3.21 ATAIL()
! รูปแบบ
ATAIL(<ชดุอาเรย>) --> คาของสมาชิกตัวสุดทาย
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สงคาของสมาชิกอาเรยตัวสุดทาย
: ตัวอยางที่ 3.31
        AR := {'C','A','F','D','B'}
        ? ATAIL(AR)                     // B
        ASORT(AR)
        ? 'MAX : ',ATAIL(AR)         // F
        ? 'MIN : ',AR[1]                // A

&  3.22 BOF()
! รูปแบบ
BOF() --> ผลการตรวจสอบเรคอรดแรกของแฟม
P หนาที่
ฟงกชนันี้จะท ําหนาที่ ตรวจสอบคาเปนจริงเมื่อถึง RECORD แรกของแฟม
: ตัวอยางที่ 3.32
        USE FILEA
        ? BOF() , RECNO()       // .F.       1
        SKIP -1
        ? BOF() , RECNO()       // .T.       1

        GO TOP
        ? BOF() , RECNO()       // .F.       1
        GO BOTTOM
        DO WHILE .NOT. BOF()      //        104       800       700
           ? FIELD1,FIELD2,FIELD3 //        103       785       200
           SKIP -1                //        102      6500       420
        ENDDO                     //        101      1000       150

&  3.23 BREAK()
! รูปแบบ
BREAK() --> NIL
P หนาที่
ฟงกชันนี้จะท ําหนาที่ หยุดการทํ างาน
: ตัวอยางที่ 3.33
        USE FILEA
        DO WHILE .NOT. EOF()
           BEGIN SEQUENCE
              ? FIELD1            //        101    ONE RECORD ONLY
              BREAK // TO END OR RECOVER
              SKIP
           END
           BREAK // TO ENDDO OF DO WHILE
        ENDDO
: ตัวอยางที่ 3.34
        USE FILEA
        DO WHILE .NOT. EOF()
           ?? '%'                         // %
           BEGIN SEQUENCE                 // 101       *%
              IF FIELD1 = '103'           // 102       *%
                 BREAK       // WOW*%
              ENDIF      // 104       *
              ? FIELD1
              SKIP      // WHEN IT SKIP, IT WILL GOTO END
           RECOVER      // IT COME FROM BREAK SO IT LOOK LIKE ELSE

              SKIP
              ? 'WOW'   // IT WILL SKIP THIS LINE
           END
           ?? '*'
        ENDDO

&  3.24 BROWSE()
! รูปแบบ
BROWSE([<มมุบน>], [<มมุซาย>], [<มุมลาง>], [<มุมขวา>]) --> NIL
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แสดงขอมูลแบบ BROWSE ในหนาตางที่กํ าหนด
: ตัวอยางที่ 3.35
        USE FILEA
        BROWSE(5,5,20,70)       // การเรียกดูขอมูลจากแฟม FILEA
        SET FILTER TO FIELD2 >= 800
        BROWSE(6,5,15,70)       // สามารถใช FILTER รวมกับ BROWSE ได

&  3.25 CDOW()
! รูปแบบ
CDOW(<นพิจนวันที>่) --> ชื่อวันในสัปดาห
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แปลงวันที ่เปนวันในสัปดาห
: ตัวอยางที่ 3.36
        ? DATE()                // 12/15/96
        ? DOW(DATE())           //  2
        ? CDOW(DATE())          // MONDAY
        ? CDOW(DATE()-3)        // SATURDAY
        ? DOW(DATE()-3)         //  7
        ? DOW(CTOD("06/04/69")) //  4
        X = "06/30/97"
        ? CDOW(CTOD(X))         // MONDAY

&  3.26 CHR()
! รูปแบบ
CHR(<จ ํานวนเลข>) --> ตัวอักษรในตารางแอสกี (ASCII)
P หนาที่
ฟงกชันนี้จะทํ าหนาที่ แปลงรหัส ASCII เปนตัวอักษร
: ตัวอยางที่ 3.37
        ? CHR(65)              // A
        ? CHR(7)                // SOUND BEEP
        ? CHR(65) + CHR(66)     // AB
        ? CHR(ASC("A")+25)      // Z
        ? "*"                   // *ABCDEFGHIJKLMNOPQRSTUVWXYZ
        FOR I = 0 TO 25
            ?? CHR(ASC("A")+I)
        NEXT

&  3.27 CMONTH()
! รูปแบบ
CMONTH(<นพิจนวนัที>่) --> ชื่อเดือนภาษาอังกฤษ
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แปลงวันที่เปนชื่อเดือน
: ตัวอยางที่ 3.38
        ? DATE()                                // 03/09/96
        ? CMONTH(DATE())+STR(YEAR(DATE()))   // MARCH 1996
        ? CMONTH(DATE()-2)                      // MARCH
        ? CMONTH(DATE()-10)                     // FEBRUARY
        ? CMONTH(DATE()-250)                    // JULY
        ? CMONTH(CTOD("06/04/69"))              // JUNE
        X = "06/30/97"
        ? CMONTH(CTOD(X))                       // JUNE
        ? SUBSTR(CMONTH(DATE()),1,3)        // MAR

&  3.28 COL()
! รูปแบบ
COL() --> ต ําแหนงหลัก

P หนาที่
ฟงกชนันี้จะท ําหนาที่ สงคาตํ าแหนงหลักปจจุบัน ไปยังตัวแปร
: ตัวอยางที่ 3.39
        ? COL()         //       0 CURRENT COLUMN =      30     20
        _COL = COL()
        _COL = _COL + 10
        @ ROW() , COL()+1 SAY "CURRENT COLUMN = "
        @ ROW() , COL()+2 SAY COL()
        @ ROW() , COL()+2 SAY _COL

&  3.29 CTOD()
! รูปแบบ
CTOD(<นพิจนอักขระ>) --> วันที่
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แปลงตัวอักษรเปนวันที่
: ตัวอยางที่ 3.40
        X = CTOD("12-31-2002")
        Y = CTOD("12/31/2002")
        ? X,Y                           // 12/31/02 12/31/02
        SET CENTURY ON
        ? X,Y                           // 12/31/2002  12/31/2002
        GDATE = DATE()
        @ ROW() , COL() SAY "GET DATE :" GET GDATE ;
                        RANGE Y , DATE()
        READ

&  3.30 CURDIR()
! รูปแบบ
CURDIR([<ชือ่ไดรฟ>]) --> ชื่อไดเร็คทรอรี่
P หนาที่
ฟงกชนันี้จะทํ าหนาที่ สงคา DIRECTORY ปจจุบัน
: ตัวอยางที่ 3.41
        ? CURDIR("C")                   // LANGUAGE\CLIPPER5
        ? CURDIR("C:")                  // LANGUAGE\CLIPPER5
        ? CURDIR("CAT")                 // LANGUAGE\CLIPPER5

        RUN CD\WINDOWS
        ? CURDIR("C")                   // WINDOWS
        RUN DIR
        RUN CD\
        ? CURDIR("C")+"*"               // *
        RUN CD\LANGUAGE\CLIPPER5
        ? CURDIR("A")                   // DATA

&  3.31 DATE()
! รูปแบบ
DATE() --> วนัที่ของเครื่อง
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สงคาวันที่ปจจุบันของเครื่อง
: ตัวอยางที่ 3.42
        ? DATE()                        // 12/31/02
        SET CENTURY ON
        ? DATE()                        // 12/31/2002

&  3.32 DAY()
! รูปแบบ
DAY(<นพิจนวันที>่) --> วันที่ของเดือน
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สงคาวันที่ของเดือน
: ตัวอยางที่ 3.43
        ? DATE()                        // 12/31/02
        ? DAY(DATE())                   // 31
        ? DAY(CTOD(""))                 // 0
        ? DAY(DATE())+1                 // 1

&  3.33 DBAPPEND()
! รูปแบบ
DBAPPEND() --> NIL
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สั่งเพิ่มเรคอรดใหม (เหมือน APPEND BLANK)

: ตัวอยางที่ 3.44
        USE FILEA
        IF RECCOUNT() > 20
           DBAPPEND()
        ELSE
           APPEND BLANK
        ENDIF

&  3.34 DBCLEARFIL()
! รูปแบบ
DBCLEARFIL() --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สั่งยกเลิกเงื่อนไขปจจุบันของฟวเตอร (เหมือน SET FILTER TO)
: ตัวอยางที่ 3.45
        USE FILEA
        SET FILTER TO FIELD2 >= 800
        COUNT TO X
        ? RECCOUNT(),X          //      4       3
        IF RECCOUNT() < 20
           SET FILTER TO
        ELSE
           DBCLEARFIL()
        ENDIF
        COUNT TO X
        ? RECCOUNT(),X          //      4       4

&  3.35 DBCLEARIND()
! รูปแบบ
DBCLEARIND() --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สั่งยกเลิกการใชแฟมดรรชนี INDEX (เหมือน SET INDEX TO)
: ตัวอยางที่ 3.46
        USE FILEA
        INDEX ON FIELD1 TO IFIELD1
        INDEX ON FIELD2 TO IFIELD2
        USE FILEA INDEX IFIELD1,IFIELD2

        LIST FIELD1,FIELD2,FIELD3
        INKEY(0)
        SET ORDER TO 2
        LIST FIELD1,FIELD2,FIELD3
        INKEY(0)
        SET INDEX TO           // ยกเลิกแฟมแบบ INDEX
        LIST FIELD1,FIELD2,FIELD3
        INKEY(0)
        SET ORDER TO 2         // ไมมีผลอีก เพราะ INDEX ถูกยกเลิกไป
        LIST FIELD1,FIELD2,FIELD3
        INKEY(0)
        DBCLEARIND()           // ยกเลิกแฟมแบบ INDEX
        LIST FIELD1,FIELD2,FIELD3

&  3.36 DBCLEARREL()
! รูปแบบ
DBCLEARREL() --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สั่งยกเลิก RELATIONS ปจจุบัน (เหมือน SET RELATION TO)
: ตัวอยางที่ 3.47
        USE FILEN INDEX IDN      // ID NAME
        USE FILES INDEX IDS NEW  // ID SALARY
        SET RELATION TO ID INTO FILEN
        LIST  ID , NAME , SALARY
        DBCLEARREL()
        LIST  ID , SALARY

&  3.37 DBCLOSEALL()
! รูปแบบ
DBCLOSEALL() --> NIL
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สั่งปดพื้นที่ท ํางานทั้งหมด (เหมือน CLOSE ALL)

: ตัวอยางที่ 3.48
        USE FILEA
        USE FILEB NEW
        LIST FIELD1,FIELD2,FIELD3
        SELE 1
        LIST FIELD1,FIELD2,FIELD3
        DBCLOSEALL()

&  3.38 DBCLOSEAREA()
! รูปแบบ
DBCLOSEAREA() --> NIL
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สั่งปดพื้นที่ท ํางานปจจุบัน (เหมือน CLOSE หรือ USE)
: ตัวอยางที่ 3.49
        USE FILEA
        USE FILEB NEW
        LIST FIELD1
        DBCLOSEAREA()
        SELE 1
        LIST FIELD1
        DBCLOSEAREA()

&  3.39 DBCOMMIT()
! รูปแบบ
DBCOMMIT() --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สั่งใหเขียนแฟมลงสื่อ มีผลอยางมากตอระบบเครือขาย (เหมือน COMMIT)
: ตัวอยางที่ 3.50
        USE FILEA
        APPEND BLANK
        REPLACE FIELD1 WITH '105',FIELD2 WITH 1250,FIELD3 WITH 100
        DBCOMMIT()

&  3.40 DBCOMMITALL()
! รูปแบบ
DBCOMMITALL() --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สั่งใหเขียนแฟมในทุกพื้นที่ลงสื่อ มีผลอยางมากตอระบบเครือขาย
: ตัวอยางที่ 3.51
        USE FILEA
        APPEND BLANK
        REPLACE FIELD1 WITH '105',FIELD2 WITH 1250,FIELD3 WITH 100
        USE FILEB NEW
        APPEND BLANK
        REPLACE FIELD1 WITH '105',FIELD2 WITH 1250,FIELD3 WITH 100
        DBCOMMITALL()

&  3.41 DBCREATE()
! รูปแบบ
DBCREATE(<ชือ่แฟมใหม>, <ชุดอาเรยเก็บโครงสราง>) --> NIL
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สั่งสรางแฟมจากโครงสรางอาเรยที่มีอยู
: ตัวอยางที่ 3.52
        FIELDAR := ARRAY(3,4)   // เตรียมใหแฟมนี้มี 3 ฟลด
        FIELDAR[1] := {'FIELD1','C',3,0}
        FIELDAR[2] := {'FIELD2','C',3,0}    ;        FIELDAR[3] := {'FIELD3','C',3,0}
        DBCREATE('TESTCRT',FIELDAR)
        USE TESTCRT
        FLDAR := ARRAY(AFIELDS())
        AFIELDS(FLDAR)
        AEVAL(FLDAR,{|X|QOUT(X)})

&  3.42 DBDELETE()
! รูปแบบ
DBDELETE() --> NIL
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สั่งลบเรคอรดปจจุบัน (เหมือน DELETE)

: ตัวอยางที่ 3.53
        USE FILEA
        DBDELETE()
        PACK

&  3.43 DBEDIT()
! รูปแบบ
DBEDIT([<มมุบน>], [<มมุซาย>], [<มุมลาง>], [<มุมขวา>],
       [<ชดุอาเรยเก็บชื่อฟลดที่ตองการแสดง>]) --> NIL
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สั่งแสดงเรคอรดที่ตองการ
: ตัวอยางที่ 3.54
        USE FILEA
        AR := {'FIELD1','FIELD2'}
        DBEDIT(5,5,20,70,AR)            // แกไขไมได
        BROWSE(4,5,15,70)               // เพิ่มลบ หรือแกไขได

&  3.44 DBF()
! รูปแบบ
DBF() --> สมนาม
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สงชื่อแฝงของพื้นที่ท ํางานปจจุบัน
: ตัวอยางที่ 3.55
        USE FILEA ALIAS NAMEFA
        ? DBF()                         // NAMEFA
        USE FILEA
        ? DBF()                        // FILEA

&  3.45 DBFILTER()
! รูปแบบ
DBFILTER() --> ขอความแสดงเงื่อนไขของตัวกรอง (FILTER)
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สงเงื่อนไขของ FILTER

: ตัวอยางที่ 3.56
        USE FILEA
        FIL := 'FIELD2 >= 800'
        SET FILTER TO &FIL
        ? DBFILTER()                    // FIELD2 > = 800

&  3.46 DBGOBOTTOM()
! รูปแบบ
DBGOBOTTOM() --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ไปยังเรคอรดสุดทาย (เหมือน GO BOTTOM)
: ตัวอยางที่ 3.57
        USE FILEA
        DBGOBOTTOM()
        WHILE !BOF()
           ? FIELD1,FIELD2,FIELD3
           SKIP -1
        END

&  3.47 DBGOTO()
! รูปแบบ
DBGOTO(<เลขทีเ่รคอรดที่ตองการไปชี>้) --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ไปยังเรคอรดที่ตองการ (เหมือน GO หรือ GOTO)
: ตัวอยางที่ 3.58
        USE FILEA
        FOR I = 1 TO RECCOUNT()
            DBGOTO(I)   ;            ? FIELD1,FIELD2,FIELD3
        NEXT

&  3.48 DBGOTOP()
! รูปแบบ
DBGOTOP() --> NIL
P หนาที่
ฟงกชนันี้จะท ําหนาที่ ไปยังเรคอรดแรก (เหมือน GO TOP)

: ตัวอยางที่ 3.59
        USE FILEA
        DBGOTO(3)
        ? FIELD1   ;        DBGOTOP()
        WHILE (!EOF())
           ? FIELD1,FIELD2,FIELD3
           DBSKIP()
        END

&  3.49 DBRECALL()
! รูปแบบ
DBRECALL() --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ยกเลิกการทํ าเครื่องหมายลบทั้งหมด (เหมือน RECALL)
: ตัวอยางที่ 3.60
        USE FILEA
        DELETE
        SKIP
        DELETE
        DBRECALL()

&  3.50 DBREINDEX()
! รูปแบบ
DBREINDEX() --> NIL
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สั่งปรับปรุงแฟมดรรชนีใหถูกตอง (เหมือน REINDEX)
: ตัวอยางที่ 3.61
        USE FILEA INDEX IFIELD1,IFIELD2
        DBREINDEX()

&  3.51 DBRELATION()
! รูปแบบ
DBRELATION(<เลขที่แฟม>) --> ชื่อฟลดของแฟมที่ระบุ
P หนาที่
ฟงกชนันี้จะท ําหนาที่ แสดงชื่อตัวแปรที่ใชเชื่อมแฟมใหสัมพันธกัน

: ตัวอยางที่ 3.62
        USE FILECUS  INDEX ICUS     // ICUS  NAMEC
        USE FILEEMP  INDEX IEMP NEW // IEMP  NAMEE
        USE FILESALE INDEX ISAL NEW // ISAL  ICUS IEMP PRODUCT
        SET RELATION ICUS INTO FILECUS,IEMP INTO FILEEMP
        ? DBRELATION(1)         // ICUS
        ? DBRELATION(2)         // IEMP

&  3.52 DBRSELECT()
! รูปแบบ
DBRSELECT(<เลขที่แฟม>) --> เลขที่พื้นที่ท ํางาน
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สงคาตัวเลขของพื้นที่ ที่มีการใชฟลดไปเชื่อมความสัมพันธมา
: ตัวอยางที่ 3.63
        USE FILECUS  INDEX ICUS     // ICUS  NAMEC
        USE FILEEMP  INDEX IEMP NEW // IEMP  NAMEE
        USE FILESALE INDEX ISAL NEW // ISAL  ICUS IEMP PRODUCT
        SET RELATION ICUS INTO FILECUS,IEMP INTO FILEEMP
        ? DBRSELECT(1) ,DBRELATION(1)     // 1     ICUS
        ? DBRSELECT(2) ,DBRELATION(2)     // 2     IEMP
        ? ALIAS(DBRSELECT(1))                 // FILECUS
        ? ALIAS(DBRSELECT(2))                 // FILEEMP

&  3.53 DBSEEK()
! รูปแบบ
DBSEEK(<ขอความที่ตองการคนหา>) --> จริงหรือเท็จ
P หนาที่
ฟงกชันนี้จะทํ าหนาที่ คนหาขอมูลตามฟลดที่ถูกจัดเรียงในแฟมดรรชนี
: ตัวอยางที่ 3.64
        USE FILEA INDEX IFIELD1
        ACCEPT TO INDF1
        IF DBSEEK(INDF1)
           ? 'FOUND' ,FIELD1,FIELD2,FIELD3
        ELSE

           ? 'NOT FOUND'
        ENDIF

&  3.54 DBSELECTAR()
! รูปแบบ
DBSELECTAR(<ชือ่พืน้ที>่ | <สมนาม>) --> NIL
P หนาที่
ฟงกชนันี้จะท ําหนาที่ เลือกพื้นที่เปาหมายตามตองการ
: ตัวอยางที่ 3.65
        USE FILEA
        USE FILEB NEW
        DBSELECTAR("FILEA")
        ? ALIAS() , SELECT()     // FILEA       1

&  3.55 DBSETFILTER()
! รูปแบบ
DBSETFILTER(<เงือ่นไขการสรางตัวกรอง>) --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ระบุเงื่อนไขในการเลือกขอมูล (เหมือน SET FILTER TO)
: ตัวอยางที่ 3.66
        USE FILEA
        DBSETFILTER({||FIELD2>=800})
        ? DBFILTER()            // ไมมีคาใหแสดง
        LIST FIELD1,FIELD2,FIELD3
        INKEY(0)
        DBSETFILTER({||FIELD2>=800},"FIELD2>=800")
        ? DBFILTER()            // FIELD2>=800
        LIST FIELD1,FIELD2,FIELD3

&  3.56 DBSETINDEX()
! รูปแบบ
DBSETINDEX(<ชือ่แฟมดรรชนี>) --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สั่งเปดแฟมดรรชนี เพื่อใหขอมูลจัดเรียงตามที่ตองการ (เหมือน SET INDEX TO)

: ตัวอยางที่ 3.67
        USE FILEA
        LIST FIELD1,FIELD2,FIELD3   // แสดงขอมูลโดยไมใชแฟมดรรชนี
        DBSETINDEX("IFIELD2")
        LIST FIELD1,FIELD2,FIELD3    // จดัเรียงตามแฟมดรรชนี IFIELD2

&  3.57 DBSETORDER()
! รูปแบบ
DBSETORDER(<เลขล ําดับของแฟมดรรชนีที่เลือก>) --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ เลือกแฟมดรรชนีที่เคยเปดไวหลายแฟม (เหมือน SET ORDER TO)
: ตัวอยางที่ 3.68
        USE FILEA INDEX IFIELD2,IFIELD3
        LIST FIELD1,FIELD2,FIELD3    // จดัเรียงตามแฟมดรรชนี IFIELD2
        DBSETORDER(2)
        LIST FIELD1,FIELD2,FIELD3    // จดัเรียงตามแฟมดรรชนี IFIELD3

&  3.58 DBSETRELAT()
! รูปแบบ
DBSETRELAT(<เลขทีพ่ืน้ที>่ | <สมนาม>, <เงื่อนไขการสรางความสัมพันธ>) --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สั่งเชื่อมพื้นที่ทํ างาน 2 พื้นที่ใหสัมพันธกัน (เหมือน SET RELATION TO)
: ตัวอยางที่ 3.69
        USE FILECUS INDEX ICUS
        USE FILEEMP INDEX IEMP NEW
        DBSETRELAT("FILECUS",{||FILEEMP->IEMP},"FILEEMP->IEMP")
        LIST  IEMP , ICUS , INAMEC , INAMEE

&  3.59 DBSKIP()
! รูปแบบ
DBSKIP([<จ ํานวนเรคอรด>]) --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ยายตัวชี้เรคอรดไปยังเรคอรดที่ตองการ (เหมือน SKIP)

: ตัวอยางที่ 3.70
        USE FILEA
        DBSKIP()
        DISPLAY FIELD1,FIELD2,FIELD3    // 2  102   6500   420
        DBSKIP(2)
        DISPLAY FIELD1,FIELD2,FIELD3    // 4  104    800   700
        DBSKIP(-1)
        DISPLAY FIELD1,FIELD2,FIELD3    // 3  103    785   200
        DBSKIP(-1)
        DISPLAY FIELD1,FIELD2,FIELD3    // 2  102   6500   420
        DBSKIP(-1)
        DISPLAY FIELD1,FIELD2,FIELD3    // 1  101   1000   150

&  3.60 DBSTRUCT()
! รูปแบบ
DBSTRUCT() --> ชดุอาเรยเก็บโครงสรางแฟม
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สราง ARRAY ที่เก็บโครงสรางฐานขอมูล
: ตัวอยางที่ 3.71
        USE FILEA
        STRUC := DBSTRUCT()
        AEVAL(STRUC,{|ELE| QOUT(ELE[1],ELE[2],ELE[3],ELE[4])})
        INKEY(0)
        AEVAL(STRUC,{|ELE| QOUT(ELE[1]) ,QQOUT(ELE[2]),;
                           QQOUT(ELE[3]) ,QQOUT(ELE[4])})

&  3.61 DBUNLOCK()
! รูปแบบ
DBUNLOCK() --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ยกเลิกการจองพื้นที่ทํ างานปจจุบัน (เหมือน UNLOCK)
: ตัวอยางที่ 3.72
        USE FILEA SHARED
        ? FLOCK()               // .T.

        ? FIELD1                // 101
        DBUNLOCK()              // DBUNLOCK() ตองใชในพื้นที่ที ่LOCK
        USE FILEA SHARED NEW
        ? FLOCK()               // .T.
        SELE 1
        ? FLOCK()               // .F.

&  3.62 DBUNLOCKALL()
! รูปแบบ
DBUNLOCKALL() --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ยกเลิกการจองพื้นที่ทํ างานในทุกพื้นที่ (เหมือน UNLOCK ALL)
: ตัวอยางที่ 3.73
        USE FILEA SHARED
          ? FLOCK()             // .T.
        USE FILEB SHARED NEW
          ? FLOCK()             // .T.
          DBUNLOCKALL()
        USE FILEA SHARED NEW
          ? FLOCK()             // .T.
        USE FILEB SHARED NEW
          ? FLOCK()             // .T.
        SELE 1
          ? FLOCK()             // .F.
        SELE 2
          ? FLOCK()             // .F.

&  3.63 DELETED()
! รูปแบบ
DELETED() --> ผลการตรวจสอบสถานะของเรคอรด
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ บอกใหทราบวาเรคอรดปจจุบัน ถูกทํ าเครื่องหมายใหถูกลบหรือไม

: ตัวอยางที่ 3.74
        USE FILEA
        ? DELETE()              // .F.
        DELETE
        ? DELETE()              // .T.
        DBRECALL()
        ? DELETE()             // .F.

&  3.64 DESCEND()
! รูปแบบ
DESCEND(<นพิจน>) --> คานิพจนเดิม
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สั่งเรียงจากมากไปนอยในแฟมดรรชนี
: ตัวอยางที่ 3.75
        USE FILEA
        INDEX ON DESCEND(FIELD3) TO IFIELD3
        LIST FIELD1,FIELD2,FIELD3
        SEEK DESCEND(200)
        IF FOUND()
           DISPLAY FIELD1,FIELD2,FIELD3 // 3   103   785  200
        ENDIF

&  3.65 DEVOUT()
! รูปแบบ
DEVOUT(<นพิจน>, [<ขอความระบุสี>]) --> NIL
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สงคาออกทางอุปกรณปจจุบัน
: ตัวอยางที่ 3.76
        USE FILEA
        WHILE !EOF()
            DEVPOS(ROW()+1,10)
            DEVOUT(FIELD1+STR(FIELD2)+STR(FIELD3),"BR+/B")
            DBSKIP()
        END

&  3.66 DEVPOS()
! รูปแบบ
DEVPOS(<แถวที>่, <หลักที>่) --> NIL
P หนาที่
ฟงกชนันี้จะท ําหนาที่ ยายตํ าแหนงจากจุดหนึ่งไปอีกจุดหนึ่ง
: ตัวอยางที่ 3.77
        USE FILEA
        I = 1
        WHILE !EOF()
            DEVPOS(I,10)
            DEVOUT(STR(I)+FIELD1+STR(FIELD2)+STR(FIELD3),"GR+/B")
            DBSKIP()
            I++
        END
: ตัวอยางที่ 3.78
        USE FILEA
        SET DEVICE TO PRINTER  // ไมเห็นผลใด ๆ ทางจอภาพ
        WHILE !EOF()
            DEVPOS(PROW()+1,10)
            DEVOUT(FIELD1+STR(FIELD2)+STR(FIELD3),"W+")
            DBSKIP()
        END
        EJECT
        SET DEVICE TO SCREEN

&  3.67 DIRECTORY()
! รูปแบบ
DIRECTORY(<กลุมแฟม>, [<ลักษณะเฉพาะ>]) --> ชุดอาเรยเก็บลักษณะแฟม
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สรางอาเรยเก็บชื่อแฟม วันที่ เวลา หรือขนาด
ตวัเลอืกของลักษณะเฉพาะ (ATTRIBUTES)
H  ชือ่แฟม รวมทั้งที่เปนแฟมซอน
S  ชือ่แฟม รวมทั้งที่เปนแฟมระบบ

D  ชือ่แฟม และชื่อ DIRECTORY ทั้งหมด
V  ชือ่อปุกรณเก็บขอมูล ซึ่งเปนชื่อ VOLUMN
ขอมูลที่ไดมา จะแบงเปน 5 สวนคือ
ชือ่แฟม ขนาด วันที่ เวลา ประเภทแฟม
: ตัวอยางที่ 3.79
        OFATTRD := DIRECTORY("C:\*.*","D")
        AEVAL(OFATTRD,{|E|QOUT  (E[1],E[2],E[3],E[4],E[5])})
        ? '---------------------------'

&  3.68 DISKSPACE()
! รูปแบบ
DISKSPACE([<เลขไดรฟ>]) --> ขนาดไดรฟ
P หนาที่
ฟงกชันนี้จะท ําหนาที่ ตรวจสอบเนื้อที่ที่เหลือในสื่อเก็บขอมูล
: ตัวอยางที่ 3.80
        ? DISKSPACE()           // 81010122  CURRENT DRIVE
        ? DISKSPACE(3)          // 81010122  DRIVE C
        ? DISKSPACE(1)          //  1153555  DRIVE A
        ? DISKSPACE(2)          //  1153555  DRIVE B ถาม ีDRIVE เดียว

&  3.69 DISPBOX()
! รูปแบบ
DISPBOX(<มมุบน>, <มุมซาย>, <มุมลาง>, <มุมขวา>,
        [<แบบของเสน>], [<สีขอบ>]) --> NIL
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แสดงกรอบสี่เหลี่ยมบนจอภาพ
ตวัเลขระบุประเภทกรอบ
1 เสนเดี่ยว
2 เสนคู
: ตัวอยางที่ 3.81
        DISPBOX(5,10,10,30,1,"BG+/B")
        DISPBOX(15,5,20,40,2,"BG+/B")

&  3.70 DISPOUT()
! รูปแบบ
DISPOUT(<นพิจน>, [<สีของตัวอักษร>]) --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ แสดงสิ่งที่ตองการทางจอภาพ โดยคํ าสั่ง SET DEVICE จะไมมีผลกับค ําสั่งนี้
: ตัวอยางที่ 3.82
        USE FILEA
        I = 1
        WHILE !EOF()
            SETPOS(I,10)
            DISPOUT(STR(I)+FIELD1+STR(FIELD2)+STR(FIELD3),"GR+/B")
            DBSKIP()   ;          I++
        END

&  3.71 DOW()
! รูปแบบ
DOW(<นพิจนวันที>่) --> วันของสัปดาห
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แปลงวันที่เปนวันของสัปดาห
: ตัวอยางที่ 3.83
        ? DATE()                // 12/15/96
        ? DOW(DATE())           //  2  (MONDAY)
        ? DOW(DATE()-3)         //  6
        ? DOW(CTOD("06/04/69")) //  4
        FOR I = DOW(DATE()) TO DOW(DATE())+2
            ? CDOW(DATE()+I)    //  TUESDAY WEDNESDAY THURSDAY
        NEXT

&  3.72 DTOC()
! รูปแบบ
DTOC(<นพิจนวันที>่) --> วันที่เปนตัวอักษร
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แปลงวันที่เปนตัวอักษร  DD/MM/YY

: ตัวอยางที่ 3.84
        ? DTOC(DATE())                  // 03/05/96
        X = DTOC(DATE())                // 03/01/96
        Y = VAL(SUBSTR(X,4,2))          // 03/02/96
        FOR I = 1 TO Y                  // 03/03/96
            ? DATE()-Y+I                // 03/04/96
        NEXT                            // 03/05/96

&  3.73 DTOS()
! รูปแบบ
DTOS(<นพิจนวันที>่) --> วันที่เปนตัวอักษร
P หนาที่
ฟงกชนันี้จะท ําหนาที่ แปลงวันที่เปนตัวอักษรในรูปแบบ YYYYMMDD
: ตัวอยางที่ 3.85
        ? DTOS(DATE())                  // 19960309
        X = DTOS(DATE())
        ? LEN(X)                        //          8
        ? "YEAR  :"+(SUBSTR(X,1,4))     // YEAR  :1996
        ? "MONTH :"+(SUBSTR(X,5,2))     // MONTH :03
        ? "DAY   :"+(SUBSTR(X,7,2))     // DAY   :09

&  3.74 EMPTY()
! รูปแบบ
EMPTY(<นพิจน>) --> ผลการตรวจสอบคาของนิพจน
P หนาที่
ฟงกชนันี้จะท ําหนาที่ ตรวจสอบคาของตัวแปรวามีคาหรือไม
เงือ่นไขการตรวจสอบส ําหรับประเภทของตัวแปร
ARRAY     : กวางเปน 0
CHARACTER : ""
NUMERIC   : 0
DATE      : NULL(CTOD(""))
LOGICAL   : FALSE (.F.)
MEMO      : ""
NIL       : NIL

: ตัวอยางที่ 3.86
        A := {}
        B := {""}
        C := " "
        D := ""
        E := 0
        F := .F.
        ? EMPTY(A)      //  .T.
        IF EMPTY(A)
           ? EMPTY(B)   //  .F.
           ? EMPTY(C)   //  .T.
           ? EMPTY(D)   //  .T.
           ? EMPTY(E)   //  .T.
           ? EMPTY(F)   //  .T.
        ENDIF

&  3.75 EOF()
! รูปแบบ
EOF() --> ผลการตรวจสอบเรคอรดสุดทายของแฟม
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ถาหมดเรคอรดสุดทายจะสงคาจริงมาให
: ตัวอยางที่ 3.87
        USE FILEA
        ? EOF() , RECNO()       // .F.       1
        GO BOTTOM
        ? EOF() , RECNO()       // .F.       4
        SKIP
        ? EOF() , RECNO()       // .T.       5
        GO TOP
        DO WHILE .NOT. EOF()      //        101      1000       150
           ? FIELD1,FIELD2,FIELD3 //        102      6500       420
           SKIP                   //        103       785       200
        ENDDO                     //        104       800       700

&  3.76 ERRORLEVEL()
! รูปแบบ
ERRORLEVEL([<คาผดิพลาด>]) --> คาผิดพลาด
P หนาที่
ฟงกชันนี้จะท ําหนาที่ ระบุคาที่จะสงกลับใหโปรแกรม
: ตัวอยางที่ 3.88
        IF !FILE("XX.PRG")
           ERRORLEVEL(25)       // TO SET VALUE OF ERRORLEVEL
           ERR = ERRORLEVEL()
        ENDIF
        IF ERRORLEVEL() = 25
           RUN DIR/S/OD/W
        ENDIF
        ? ERR                   //     25

&  3.77 EVAL()
! รูปแบบ
EVAL(<นพิจนค ํานวณ>, [<รายการคาตาง ๆ ที่ปอนใหนิพจน>]) --> คาที่ไดจากนิพจน
P หนาที่
ฟงกชันนี้จะทํ าหนาที่ ตรวจสอบโปรแกรม
: ตัวอยางที่ 3.89
        BCODE = {|A|A+1}
        ? EVAL(BCODE,5)         //     6
        ? EVAL(BCODE,2)         //     3
        ? EVAL(BCODE,7)         //     8
        ? (BCODE,5)             //     5
        ? BCODE                 //    NOTHING
: ตัวอยางที่ 3.90
        BCODE = &("{|A|A*5-2}")
        FOR I = 1 TO 3          //     3
           ? EVAL(BCODE,I)      //     8
        NEXT                    //    13

: ตัวอยางที่ 3.91
        GRADE = {|G|IF(G>80,'A',(IF(G>70,'B','F')))}
        ? EVAL(GRADE,85)        // A
        ? EVAL(GRADE,75)        // B
        ? EVAL(GRADE,65)        // F

&  3.78 EXP()
! รูปแบบ
EXP(<นพิจนตวัเลข>) --> ผลการยกกํ าลังของคา E
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สั่งคํ านวณ E ยกกํ าลัง X
: ตัวอยางที่ 3.92
        ? EXP(1)                //      2.72
        SET DECIMAL TO 10
        ? EXP(1)                //      2.7182818285
        ? LOG(EXP(5))           //      5.0000000000
        ? EXP(LOG(5))           //      5.0000000000

&  3.79 FCLOSE()
! รูปแบบ
FCLOSE(<ชือ่ตวัแปรระดับแฟม>) --> ผลการปดแฟม
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สั่งปดแฟมที่เปดแบบ BINARY แลว น ําคาใน BUFFER เขียนลง DISK
: ตัวอยางที่ 3.93
        F := FCREATE("TEST.TXT",0)  // เปดแฟมแบบอานเขียนได
        FWRITE(F,"COMPUTER",10)     // แฟมจะมีขนาด 10 ไบต
        FCLOSE(F)

&  3.80 FCOUNT()
! รูปแบบ
FCOUNT() --> จ ํานวนฟลดของแฟม
P หนาที่
ฟงกชันนี้จะท ําหนาที่ จะแสดงจํ านวณฟลดของแฟม

: ตัวอยางที่ 3.94
        USE FILEA
        AFLDNAME := ARRAY(AFIELDS())
        AFLDTYPE := ARRAY(AFIELDS())
        AFLDWIDTH:= ARRAY(AFIELDS())
        AFLDDEC  := ARRAY(AFIELDS())
        AFIELDS(AFLDNAME,AFLDTYPE,AFLDWIDTH,AFLDDEC)
        FOR I = 1 TO AFIELDS()                  // FIELD1 C  10
            ?  AFLDNAME[I],AFLDTYPE[I]          // FIELD2 N  10  0
            ?? STR(AFLDWIDTH[I],10)             // FIELD3 N  10  0
            ?? IF(AFLDTYPE[I]='N',STR(AFLDDEC[I],10),'')
        NEXT
        ? AFIELDS()     //     3
        ? FCOUNT()      //     3

&  3.81  FCREATE()
! รูปแบบ
FCREATE(<ชือ่แฟม>, [<เลขระบุลักษณะเฉพาะ>]) --> ตัวแปรระดับแฟม
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สรางแฟมแบบ BINARY FILE
การสรางแฟมสามารถระบุแฟมไดหลายแบบดังนี้
0  อาน/เขียนได
1  อานอยางเดียว
2  แฟมแบบซอน
3  แฟมระบบ
: ตัวอยางที่ 3.95
        F := FCREATE("TEST.TXT",0)  // เปดแฟมแบบอานเขียนได
        FWRITE(F,"AA",10)
        FWRITE(F,"AA",10)
        FWRITE(F,"AA",10)
        FWRITE(F,"AA",10)
        FCLOSE(F)                   // แฟมจะมีขนาด 40 ไบต
        ? FERROR()                  // 0 แสดงวาปดแฟมไดไมผิดพลาด

&  3.82 FERASE()
! รูปแบบ
FERASE(<ชือ่แฟม>) --> ผลการลบแฟม
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สั่งลบแฟมจากสื่อเก็บขอมูล
: ตัวอยางที่ 3.96
        IF FERASE("TEST.TXT") = -1
           ? "ERROR ON ERASING"
        ENDIF

&  3.83 FERROR()
! รูปแบบ
FERROR() --> เลขระบุความผิดพลาด
P หนาที่
ฟงกชนันี้จะท ําหนาที่ ตรวจสอบขอผิดพลาดของการทํ างานกับ BINARY FILE
: ตัวอยางที่ 3.97
        F := FCREATE("TEST.TXT",0)  // เปดแฟมแบบอานเขียนได
        IF FERROR() = 0
           FWRITE(F,"AA",10)
           FWRITE(F,"AA",10)
           FWRITE(F,"AA",10)
           FWRITE(F,"AA",10)
           FCLOSE(F)                // แฟมจะมีขนาด 40 ไบต
           ? FERROR()               // 0 แสดงวาปดแฟมไดไมผิดพลาด
        ENDIF

&  3.84 FIELDNAME()
! รูปแบบ
FIELDNAME(<ต ําแหนงฟลด>) --> ชื่อฟลด
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แสดงชื่อฟลดจากแฟมปจจุบัน
: ตัวอยางที่ 3.98
        USE FILEA
        ? FIELDNAME(1)          // FIELD1 แสดงชื่อฟลดแรกของแฟม
        FOR I = 1 TO FCOUNT()

            ? FIELDNAME(I)
        NEXT

&  3.85 FIELDGET()
! รูปแบบ
FIELDGET(<ต ําแหนงฟลด>) --> คาของฟลด
P หนาที่
ฟงกชันนี้จะท ําหนาที่ รับคาจากจากฟลด
: ตัวอยางที่ 3.99
        USE FILEA
        ? FIELD1                // 101
        ? FIELD2                //       1000
        ? FIELDGET(1)           // 101
        ? FIELDGET(2)           //       1000
        X := FIELDNAME(1)
        Y := &X
        ? Y                     // 101

&  3.86 FIELDPOS()
! รูปแบบ
FIELDPOS(<ชือ่ฟลด>) --> ตํ าแหนงฟลด
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แสดงตํ าแหนงของฟลดปจจุบัน
: ตัวอยางที่ 3.100
        USE FILEA
        ? FIELDPOS("FIELD1")             //        1
        ? FIELDPOS("FIELD2")             //        2
        ? FIELDGET(FIELDPOS("FIELD1")) // 101

&  3.87 FIELDPUT()
! รูปแบบ
FIELDPUT(<ต ําแหนงฟลด>, <คาใหม>) --> คาใหมที่แทนที่คาเกาในฟลด
P หนาที่
ฟงกชันนี้จะท ําหนาที่ ปรับปรุงคาในฟลด

: ตัวอยางที่ 3.101
        USE FILEA
        ? FIELDGET(1)       // 101
        FIELDPUT(1,"103")   // ท ําใหคาในฟลดเปลี่ยนไปโดยไมตอง REPLACE
        ? FIELDGET(1)       // 103

&  3.88 FILE()
! รูปแบบ
FILE(<ชือ่แฟม>) --> ผลการตรวจสอบการเปดแฟม
P หนาที่
ฟงกชันนี้จะท ําหนาที่ ตรวจสอบการเปดแฟม
: ตัวอยางที่ 3.102
        ? FILE("COMMAND.COM")           // .F.
        ? FILE("\WINDOWS\WIN.COM")      // .T.
        SET PATH TO \WINDOWS
        ? FILE("SYSTEM.INI")            // .T.
        SET DEFAULT TO C:\DOS
        ? FILE("COMMAND.COM")           // .T.
        SET PATH TO \WINDOWS
        SET DEFAULT TO C:\DOS
        ? FILE("XCOPY.EXE")             // .T.
        ? FILE("WIN.INI")               // .T.
        ? CURDIR("C:")                  // LANGUAGE\CLIPPER5

